Examples of computing Ideal Quotients $I : J$

```
> interface(imaginaryunit=_i): # so we can use I for ideals
> with(PolynomialIdeals):
> I, J, K := <x*z,y*z>, <x,y>, <z>;
```

$$I, J, K := \langle x\,z, y\,z\rangle, \langle x, y\rangle, \langle z\rangle$$

Check Proposition 9 (i): $I : \langle 1\rangle = I$

```
> Quotient(I,<1>);
```

$$\langle x\,z, y\,z\rangle$$

Check Proposition 9 (iii): If $J \subseteq I$ then $I:J = \langle 1\rangle$

```
> Quotient(I,<x*z>);
```

$$\langle 1\rangle$$

To compute I : J where $J = \langle x\rangle + \langle y\rangle$ use Proposition 10 (3): I:J = I:$\langle x\rangle \cap I:\langle y\rangle$ and compute these using Theorem 11.

```
> Quotient(I,<x>), Quotient(I,<y>);
```

$$\langle z\rangle, \langle z\rangle$$

Thus I : $\langle x\rangle$ = $\langle z\rangle$ and $I:\langle y\rangle$ = $\langle z\rangle$ and their intersection is clearly $\langle z\rangle$. Check using Maple

```
> Quotient(I,J);
```

$$\langle z\rangle$$

```
> J := [x,y];
```

$$J := [x, y]$$

```
> K := [y^2-x*z,x^3-y*z,x^2*y-z^2];
```

$$K := [-x\,z + y^2, x^3 - y\,z, x^2\,y - z^2]$$

Construct $I = J \cap K$ using the algorithm from section 4.3: $J \cap K = t\,J + (1-t)\,K \cap k[x, y, z]$

```
> I := Groebner[Basis]([seq(t*f,f=J),seq((1-t)*g,g=K)], plex(t,x,y,z));
```

$$I := [y^6 - y\,z^4, x\,z - y^2, x\,y^4 - y\,z^3, x^2\,y^2 - y\,z^2, x^3 - y\,z, t\,z^2 + x^2\,y - z^2, t\,y, t\,x]$$

```
> I := remove(has,I,t);
```

$$I := [y^6 - y\,z^4, x\,z - y^2, x\,y^4 - y\,z^3, x^2\,y^2 - y\,z^2, x^3 - y\,z]$$

```
> I, J, K := <I>, <J>, <K>;
```

$$I, J, K := \langle y^6 - y\,z^4, x\,z - y^2, x^3 - y\,z, x^2\,y^2 - y\,z^2, x\,y^4 - y\,z^3\rangle, \langle x, y\rangle, \langle -x\,z + y^2, x^2\,y - z^2, x^3 - y\,z\rangle$$

```
> L := Intersect(J,K);
```

$$L := \langle -x\,z + y^2, x^3 - y\,z\rangle$$

The basis we computed is different from the one Intersect( ... ) computed. The difference is just

```
> Groebner[Basis]( I, tdeg(x,y,z) );
```

$$[-x\,z + y^2, x^3 - y\,z]$$

```
> Quotient( I, K );
```

$$\langle x, y\rangle$$

To compute I : K using Proposition 9 (3) and Theorem 11, this time $K = \langle g1\rangle + \langle g2\rangle + \langle g3\rangle$ .

```
> g := Generators(K);
```

$$g := \{-x\,z + y^2,\ x^2\,y - z^2,\ x^3 - y\,z\}$$

```
> T1, T2, T3 := seq( Intersect(I,<g[i]>), i=1..3 );
```

$$T1,\ T2,\ T3 := \langle x\,z - y^2\rangle,\ \langle x^2\,y^2 - y\,z^2,\ x^3\,y - x\,z^2\rangle,\ \langle x^3 - y\,z\rangle$$

From this we see that a basis for I : <$g_1$> is <1> and similarly I : <$g_3$> = $\langle 1\rangle$.

```
> < seq( normal(f/g[2]), f=Generators(T2) )>;
```

$$\langle x,\ y\rangle$$

Thus *I : K* = $\langle 1\rangle \cap \langle x, y\rangle = \langle x, y\rangle$ .